

6.4 One Householder Transformation

A. Purpose

These subroutines compute the parameters defining a Householder orthogonal transformation which zeroes specified components in a given vector, or in a column or row of a matrix. Also they optionally apply either a newly defined transformation or a previously defined Householder transformation to a vector or a set of vectors, which typically would be columns or rows of a matrix.

Two versions are provided. For the usual case in which both the pivot vector and the vectors to which the transformation is to be applied are column vectors, we recommend the use of DHTCC or SHTCC. If either the pivot vector or the vectors to which the transformations are to be applied, or both, are row vectors, one must use DHTGEN or SHTGEN.

These subroutines are used by other library subroutines, and can be used as modules in implementing other linear algebra algorithms.

B. Usage

B.1 Usage of SHTGEN and DHTGEN

We use the term *pivot vector* to mean the vector that plays a special role in determining or defining the Householder transformation. When $\text{MODE} = 1$ a transformation will be determined that is appropriate to zero components L1 through M of the pivot vector. These components are then replaced by values that partially define the appropriate Householder transformation. Component LPIVOT of the pivot vector, and UPARAM are also assigned values to complete the definition of the transformation. After defining the transformation it will optionally be applied to specified vectors.

When $\text{MODE} = 2$ the transformation defined in a previous call with $\text{MODE} = 1$ will be applied to specified vectors.

B.1.a Program Prototype, Single Precision

INTEGER MODE, LPIVOT, L1, M, LDU,
LDC, NCV

LOGICAL COLU, COLC

REAL U(LDU,*), UPARAM, C(LDC,*)

Assign values to all integer and logical arguments and to the relevant real arguments.

```
CALL SHTGEN(MODE, LPIVOT, L1,  
            M, U, LDU, COLU, UPARAM,  
            C, LDC, NCV, COLC)
```

©1997 Calif. Inst. of Technology, 2015 Math à la Carte, Inc.

Computed results will be returned in U(.), UPARAM, and/or C(.), depending on the initial settings of the integer arguments.

B.1.b Argument Definitions

MODE [in] Set by the user to the value 1 or 2. If $\text{MODE} = 1$, the subroutine computes parameters defining a Householder transformation as described in Section D. If $\text{MODE} = 2$, it is assumed that Householder transformation parameters have already been defined by a previous call with $\text{MODE} = 1$. In either case, if $\text{NCV} > 0$, the subroutine will apply the current transformation to the set of NCV M-vectors stored in the array C(.).

LPIVOT, L1, M [in] These integers specify that the vectors to be referenced or operated upon are M-dimensional, and the components to be referenced or operated upon in each vector are the one indexed by LPIVOT and the set indexed from L1 through M. To define a nontrivial transformation these integers must satisfy

$$1 \leq \text{LPIVOT} < \text{L1} \leq \text{M}$$

If these inequalities are not all satisfied, the subroutine returns without doing any computation. This is not regarded as an error. It has the implicit effect of applying an identity transformation.

U(.) [inout] The array U(.) contains the pivot vector as either a column or a row. Examples:

If the pivot vector is column J of an array declared as A(IDIM,JDIM), the arguments corresponding to “U, LDU, COLU” should be written as “A(1, J), IDIM, .true.”.

If the pivot vector is row I of an array declared as A(IDIM,JDIM), the arguments corresponding to “U, LDU, COLU” should be written as “A(I,1), IDIM, .false.”.

LDU [in] Leading dimensioning parameter for U(.).

COLU [in] Must be set to .true. if the pivot vector is a column of U(.) and to .false. if the pivot vector is a row of U(.).

UPARAM [inout] Holds a value completing the definition of a Householder transformation. See description in Section D. This value will be computed by the subroutine when $\text{MODE} = 1$ and must be present on entry when $\text{MODE} = 2$.

C(,) [inout] If $NCV \leq 0$, no reference will be made to $C(,)$. If $NCV > 0$, the array $C(,)$ is regarded as containing a set of NCV M -vectors. These are regarded as column vectors if $COLC = .true.$ and row vectors if $COLC = .false.$ On entry $C(,)$ contains vectors to be transformed, and on return $C(,)$ contains the vectors resulting from application of the current Householder transformation. Examples:

To apply a transformation to columns $J + 1$ through N of an array declared as $A(IDIM,JDIM)$, the arguments “ $C, LDC, NCV, COLC$ ” should be written as “ $A(1, J+1), IDIM, N-J, .true.$ ”.

To apply a transformation to rows $K + 1$ through $M1$ of an array declared as $A(IDIM,JDIM)$, the arguments “ $C, LDC, NCV, COLC$ ” should be written as “ $A(K+1, 1), IDIM, M1-K, .false.$ ”.

LDC [in] Leading dimensioning parameter of $C(,)$.

NCV [in] Number of vectors in $C(,)$ to be transformed. If $NCV \leq 0$, the array $C(,)$ will not be referenced.

COLC [in] Must be set to $.true.$ if the vectors to be transformed are column vectors of $C(,)$, and to $.false.$ if the vectors to be transformed are row vectors of $C(,)$.

B.2 Usage of SHTCC and DHTCC

Subroutine $DHTCC$ is a modification of $DHTGEN$ specialized for the case of $COLU = .true.$ and $COLC = .true.$

B.2.a Program Prototype, Single Precision

INTEGER MODE, LPIVOT, L1, M, LDC, NCV

REAL U(*), UPARAM, C(LDC,*)

Assign values to all integer and logical arguments and to the relevant real arguments.

CALL SHTCC(MODE, LPIVOT, L1, M, U, UPARAM, C, LDC, NCV)

Computed results will be returned in $U(,)$, $UPARAM$, and/or $C(,)$, depending on the initial settings of the integer arguments.

B.2.b Argument Definitions

The arguments have the same meanings as the arguments of the same names in the call to $SHTGEN$. Note that arguments LDU , $COLU$, and $COLC$ are not used, and U is a one-dimensional array. This subroutine functions as though $COLU = .true.$ and $COLC = .true.$

B.3 Modifications for Double Precision

For double precision usage change the $REAL$ statement to $DOUBLE PRECISION$ and change the subroutine names $SHTGEN$ and $SHTCC$ to $DHTGEN$ and $DHTCC$ respectively.

C. Examples and Remarks

The program $DRSHTCC$ illustrates the use of $SHTCC$. The array $D(,)$ is initialized with a 3×2 matrix, A , in its first 2 columns, and a 3×3 identity matrix in columns 3 through 5. Subroutine $SHTCC$ is used to compute the QR factorization of A , *i.e.*, the program determines an orthogonal matrix, Q , and a triangular matrix, R , such that $A = QR$. Results are shown in $ODSHTCC$.

For additional examples of the use of $SHTCC$ and $SHTGEN$ see the source code for the library subroutines $SHFTI$ or $SSVDRS$.

D. Functional Description

Let \mathbf{u} be a nonzero n -vector and define $\beta = -\|\mathbf{u}\|^2/2$. The matrix

$$Q = I + \beta^{-1}\mathbf{u}\mathbf{u}^t$$

is symmetric and orthogonal, and is called a Householder transformation matrix.

A product of the form $\mathbf{w} = Q\mathbf{v}$ can be computed by the steps:

$$\gamma = (\mathbf{u}^t\mathbf{v})/\beta$$

$$\mathbf{w} = \mathbf{v} + \gamma\mathbf{u}$$

Typically one desires a \mathbf{u} such that for some particular nonzero vector, \mathbf{v} , the vector $\mathbf{w} = Q\mathbf{v}$ will consist of zeros except for the first component. Given \mathbf{v} , the appropriate \mathbf{u} and β to accomplish this can be defined mathematically as follows:

$$\sigma = 1 \quad \text{if } v_1 > 0 \text{ and } -1 \text{ otherwise}$$

$$u_1 = v_1 + \sigma\|\mathbf{v}\|$$

$$u_i = v_i, \quad i = 2, \dots, n$$

$$\beta = -\sigma u_1\|\mathbf{v}\| \quad (\text{Note that } \beta \leq 0.)$$

With this definition of σ , \mathbf{u} , and β , and thus of Q , the vector $\mathbf{w} = Q\mathbf{v}$ will have components:

$$w_1 = -\sigma\|\mathbf{v}\|$$

$$w_i = 0, \quad i = 2, \dots, n.$$

In applications it is frequently convenient to regard the vectors \mathbf{v} , \mathbf{u} , and \mathbf{w} of the above formulas as embedded in higher dimensional vectors with other components which play no role in the current transformation. Specifically we regard the host vector for \mathbf{v} (and similarly for \mathbf{u} and \mathbf{w}) as being of dimension M , with v_1 at position $LPIVOT$ and v_2, \dots, v_n in positions $L1$ through M . Thus we are

identifying n with the value $M - L1 + 2$. We assume $1 \leq \text{LPIVOT} < L1 \leq M$.

On input with $\text{MODE} = 1$ the array $U(,)$ contains \mathbf{v} embedded as specified by the integers LPIVOT , $L1$, and M , and with the Fortran 77 storage mapping as specified by the arguments LDU and COLU . The subroutine computes w_1 and \mathbf{u} . It stores the u_1 in UPARAM , w_1 in place of v_1 , and u_2, \dots, u_M in place of v_2, \dots, v_M .

When applying transformations the quantity β is computed as $\beta = u_1 w_1$, using the value of u_1 stored in UPARAM and w_1 stored in $U(,)$.

This is essentially Algorithm H12 of [1] with a change in the way the user specifies the column/row options.

References

1. Charles L. Lawson and Richard J. Hanson, **Solving Least-Squares Problems**, Prentice-Hall, Englewood Cliffs, N. J. (1974) 340 pages.

E. Error Procedures and Restrictions

To define and/or apply a nontrivial transformation one must have

$$1 \leq \text{LPIVOT} < L1 \leq M$$

If these conditions are not all satisfied the subroutine returns immediately, doing no computation. This is not regarded as an error condition as it may occur intentionally as an end condition in a loop. It will have the effect of computing and/or applying an identity transformation.

If, on entry with $\text{MODE} = 1$, the components of the pivot vector indexed by LPIVOT , and $L1$ through M , are all zero, the subroutine will effectively define an identity transformation. It will set $\text{UPARAM} = 0$. This is not regarded as an error condition.

F. Supporting Information

The source language is ANSI Fortran 77.

Entry	Required Files
DHTCC	DHTCC, DNRM2
DHTGEN	DAXPY, DDOT, DHTGEN, DNRM2
SHTCC	SHTCC, SNRM2
SHTGEN	SAXPY, SDOT, SHTGEN, SNRM2

Original version designed and programmed by C. L. Lawson and R. J. Hanson, JPL, 1968, and published in [1] as subroutine H12. Adapted to Fortran 77 for the MATH77 library by Lawson and S. Y. Chiu, June 1987.

DRSHTCC

```
c   program DRSHTCC
c>> 1994-10-19 DRSHTCC Krogh Changes to use M77CON
c>> 1987-12-09 DRSHTCC Lawson Initial Code.
c   Demo driver for SHTCC, Householder transformations.
c—S replaces "?": DR?HTCC, ?HTCC, ?MATP
      integer IDIM, JDIM, J, M, N
      parameter(IDIM = 3, JDIM = 5, M = 3, N = 2)
      real      D(IDIM,JDIM), UPARAM, ZERO
      parameter(ZERO = 0.0E0)
      data (D(1,J),J=1,5) / 0.870E0, 0.796E0, 1.0E0, 0.0E0, 0.0E0 /
      data (D(2,J),J=1,5) / 0.571E0, -0.804E0, 0.0E0, 1.0E0, 0.0E0 /
      data (D(3,J),J=1,5) / -0.960E0, 0.346E0, 0.0E0, 0.0E0, 1.0E0 /
c
      print*, 'DRSHTCC.. Demonstrate computation of a QR decomposition.'
      print*, 'Given matrix A, compute orthogonal Q and triangular R'
      print*, 'such that A = Q * R.'
      call SMATP(D,IDIM,M,N, '0 A =')
      do 20002 J=1,N
         call SHTCC(1,J,J+1,M,D(1,J),UPARAM,D(1,J+1),IDIM,M+N-J)
20002 continue
      call SMATP(D(1,N+1),IDIM,M,M, '0 Q [Transposed] =')
      D(2,1) = ZERO
      D(3,1) = ZERO
      D(3,2) = ZERO
      call SMATP(D,IDIM,M,N, '0 R =')
      stop
      end
```

ODSHTCC

DRSHTCC.. Demonstrate computation of a QR decomposition.
Given matrix A, compute orthogonal Q and triangular R
such that $A = Q * R$.

A =

		COL 1	COL 2
ROW	1	0.8700000	0.7960000
ROW	2	0.5710000	-0.8040000
ROW	3	-0.9600000	0.3460000

Q [Transposed] =

		COL 1	COL 2	COL 3
ROW	1	-0.6144857	-0.4033004	0.6780532
ROW	2	0.7102542	-0.6569378	0.2529267
ROW	3	0.3434333	0.6370100	0.6901246

R =

		COL 1	COL 2
ROW	1	-1.415818	6.9729328E-02
ROW	2	0.000000	1.181053
ROW	3	0.000000	0.000000